

## COMPUTATIONS USING THEOREM 6.4

LINDSAY MARJANSKI, VINCENT SOLON, FRANK ZHENG, KATHLEEN ZOPFF

ABSTRACT. We prove the results of [1] and [2] using Theorem 6.4. We also provide a Sage program that automatically carries out these computations.

### 1. STATEMENT OF THEOREM 6.4

**Theorem 6.4.** Let  $\Gamma$  be a link-regular simplicial graph with maximum clique size  $d$ . Let  $\ell_0 = |V(\Gamma)|$ , and for  $1 \leq k \leq d$ , let  $\ell_k = |\text{Lk}(\sigma)|$ , where  $\sigma$  is any  $k$ -clique (this is well-defined by link-regularity of  $\Gamma$ ). For integers  $0 \leq m \leq d$  and  $0 \leq k \leq m$ , set

$$N_{m,k} = \begin{cases} \left( \prod_{k < j < m} \ell_j \right) \sum_{j=k}^m \binom{m-k}{j-k} (-1)^{j-k} \ell_j & \text{if } k < m-1 \\ \ell_{m-1} - \ell_m - 1 & \text{if } k = m-1 \\ 1 & \text{if } k = m, \end{cases}$$

and for integers  $0 \leq i \leq d$  and  $0 \leq j \leq i$ , set

$$\begin{cases} M_{i,j} = \sum_{\substack{j \leq s_1 < t_1 \leq \dots \leq s_n < t_n \leq d \\ (t_1 - s_1) + \dots + (t_n - s_n) = i - j}} (-1)^n \binom{t_1}{s_1} \dots \binom{t_n}{s_n} N_{t_1, s_1} \dots N_{t_n, s_n} & \text{if } i \neq j \\ M_{i,j} = 1 & \text{if } i = j. \end{cases}$$

Then the geodesic growth of  $\text{RACG}(\Gamma)$  is given by the rational function

$$\mathcal{G}(z) = \frac{\sum_{i=0}^d \left( \sum_{j=0}^i \ell_0 \dots \ell_{j-1} M_{i,j} \right) z^i}{\sum_{i=0}^d M_{i,0} z^i}.$$

### 2. CALCULATION WHEN $d = 2$

First, we prove the Theorem 5.1 in [1].

**Theorem 2.1.** Let  $\Gamma$  be an  $\ell$ -regular triangle-free graph with  $n$  vertices. The geodesic growth of  $\text{RACG}(\Gamma)$  is given by the rational function

$$\mathcal{G}(z) = \frac{1 - (\ell - 3)z + 2z^2}{1 + (-n - \ell + 3)z + (-2n + 2 + n\ell)z^2}.$$

*Proof.* We apply Theorem 6.4 in the case  $d = 2$ . Note that  $\ell_0 = n$ ,  $\ell_1 = \ell$ , and  $\ell_2 = 0$ . For the  $N_{m,k}$ , we have

$$\begin{aligned} N_{m,k} &= 1 && \text{if } m = k, \\ N_{1,0} &= \ell_0 - \ell_1 - 1 = n - \ell - 1 \\ N_{2,1} &= \ell_1 - \ell_2 - 1 = \ell - 1 \\ N_{2,0} &= \ell_1 \sum_{j=0}^2 \binom{2}{j} (-1)^j \ell_j = \ell(n - 2\ell). \end{aligned}$$

Next, we compute the  $M_{i,j}$ . We have  $M_{i,j} = 1$  when  $i = j$  by definition, so we only need to consider when  $i \neq j$ . Let us first work through the computation of  $M_{2,0}$  in detail. By definition, we have

$$M_{2,0} = \sum_{\substack{0 \leq s_1 < t_1 \leq \dots \leq s_n < t_n \leq 2 \\ (t_1 - s_1) + \dots + (t_n - s_n) = 2}} (-1)^n \binom{t_1}{s_1} \cdots \binom{t_n}{s_n} N_{t_1, s_1} \cdots N_{t_n, s_n}$$

The “partitions”  $(s_1 < t_1) \leq \dots \leq (s_n < t_n)$  that this sum ranges over are  $(0 < 1) \leq (1 < 2)$  and  $(0 < 2)$ . Thus,

$$\begin{aligned} M_{2,0} &= (-1)^2 \binom{1}{0} \binom{2}{1} N_{1,0} N_{2,1} + (-1)^1 \binom{2}{0} N_{2,0} \\ &= 2(n - \ell - 1)(\ell - 1) - \ell(n - 2\ell) \\ &= n\ell - 2n + 2. \end{aligned}$$

The other  $M_{i,j}$  are computed similarly:

$$\begin{aligned} M_{1,0} &= (-1)^1 \binom{1}{0} N_{1,0} + (-1)^1 \binom{2}{1} N_{2,1} = -n - \ell + 3 \\ M_{2,1} &= (-1)^1 \binom{2}{1} N_{2,1} = -2\ell + 2. \end{aligned}$$

Finally, the numerator of the geodesic growth is computed as

$$M_{0,0} + (M_{1,0} + nM_{1,1})z + (M_{2,0} + nM_{2,1} + n\ell M_{2,2})z^2 = 1 + (-\ell + 3)z + 2z^2.$$

and the denominator is

$$M_{0,0} + M_{1,0}z + M_{2,0}z^2 = 1 + (-n - \ell + 3)z + (n\ell - 2n + 2)z^2,$$

which completes the proof.  $\square$

### 3. CALCULATION WHEN $d = 3$

Next, we prove the main result of [2]. (Note that in [2], they give a formula of  $\mathcal{G}(z) - 1$ , so the numerator of the below rational function is slightly different.)

**Theorem 3.1.** Let  $\Gamma$  be a link-regular graph with  $n$  vertices and no 4-cliques, and let  $\ell = \ell_1$  and  $q = \ell_2$  (where  $\ell_1, \ell_2$  are as in Theorem 6.4). Then the geodesic growth of  $\text{RACG}(\Gamma)$  is given by the rational function

$$\frac{1 + (-\ell - q + 6)z + (\ell q - 3\ell - q + 11)z^2 + 6z^3}{1 + (-n - \ell - q + 1)z + (n\ell + nq + \ell q - 5n - 3\ell - q + 11)z^2 + (-n\ell q + 3n\ell - 6n + 6)z^3}$$

*Proof.* We apply Theorem 6.4 in the case  $d = 3$ . For the  $N_{m,k}$ , we have

$$\begin{aligned} N_{m,k} &= 1 && \text{if } m = k, \\ N_{1,0} &= \ell_0 - \ell_1 - 1 = n - \ell - 1 \\ N_{2,1} &= \ell_1 - \ell_2 - 1 = \ell - q - 1 \\ N_{2,0} &= \ell_1 \sum_{j=0}^2 \binom{2}{j} (-1)^j \ell_j = \ell(n - 2\ell + q) \\ N_{3,2} &= \ell_2 - \ell_3 - 1 = q - 1 \\ N_{3,1} &= \ell_2 \sum_{j=1}^3 \binom{2}{j-1} (-1)^{j-1} \ell_j = q(\ell - 2q) \\ N_{3,0} &= \ell_1 \ell_2 \sum_{j=0}^3 \binom{3}{j} (-1)^j \ell_j = \ell q(n - 3\ell + 3q). \end{aligned}$$

For the  $M_{i,j}$ , we have

$$\begin{aligned} M_{i,j} &= 1 && \text{if } i = j, \\ M_{1,0} &= (-1)^1 \binom{1}{0} N_{1,0} + (-1)^1 \binom{2}{1} N_{2,1} + (-1)^1 \binom{3}{2} N_{3,2} \\ &= -(n - \ell - 1) - 2(\ell - q - 1) - 3(q - 1) \\ &= -n - \ell - q + 6 \\ M_{2,0} &= (-1)^2 \binom{1}{0} \binom{2}{1} N_{1,0} N_{2,1} + (-1)^2 \binom{1}{0} \binom{3}{2} N_{1,0} N_{3,2} \\ &\quad + (-1)^2 \binom{2}{1} \binom{3}{2} N_{2,1} N_{3,2} + (-1)^1 \binom{2}{0} N_{2,0} + (-1)^1 \binom{3}{1} N_{3,1} \\ &= 2(n - \ell - 1)(\ell - q - 1) + 3(n - \ell - 1)(q - 1) + 6(\ell - q - 1)(q - 1) \\ &\quad - \ell(n - 2\ell + q) - 3q(\ell - 2q) \\ &= n\ell + nq + \ell q - 5n - 3\ell - q + 11 \\ M_{2,1} &= (-1)^1 \binom{2}{1} N_{2,1} + (-1)^1 \binom{3}{2} N_{3,2} \\ &= -2(\ell - q - 1) - 3(q - 1) \\ &= -2\ell - q + 5 \end{aligned}$$

$$\begin{aligned}
M_{3,0} &= (-1)^3 \binom{1}{0} \binom{2}{1} \binom{3}{2} N_{1,0} N_{2,1} N_{3,2} + (-1)^2 \binom{1}{0} \binom{3}{1} N_{1,0} N_{3,1} \\
&\quad + (-1)^2 \binom{2}{0} \binom{3}{2} N_{2,0} N_{3,1} + (-1)^1 \binom{3}{0} N_{3,0} \\
&= -6(n - \ell - 1)(\ell - q - 1)(q - 1) + 3(n - \ell - 1)(q(\ell - 2q)) \\
&\quad + 3\ell(n - 2\ell + q)(q - 1) - \ell q(n - 3\ell + 3q) \\
&= -n\ell q + 3n\ell - 6n + 6 \\
M_{3,1} &= (-1)^2 \binom{2}{1} \binom{3}{2} N_{2,1} N_{3,2} + (-1)^1 \binom{3}{1} N_{3,1} \\
&= 6(\ell - q - 1)(q - 1) + 3q(\ell - 2q) \\
&= 3\ell q - 6\ell + 6 \\
M_{3,2} &= (-1)^1 \binom{3}{2} N_{3,2} \\
&= -3q + 3.
\end{aligned}$$

Finally, the numerator of the geodesic growth is computed as

$$\begin{aligned}
&M_{0,0} + (M_{1,0} + nM_{1,1})z + (M_{2,0} + nM_{2,1} + n\ell M_{2,2})z^2 \\
&\quad + (M_{3,0} + nM_{3,1} + n\ell M_{3,2} + n\ell q M_{3,3})z^3 \\
&= 1 + (-\ell - q + 6)z + (\ell q - 3\ell - q + 11)z^2 + 6z^3.
\end{aligned}$$

and the denominator is

$$\begin{aligned}
&M_{0,0} + M_{1,0}z + M_{2,0}z^2 + M_{3,0}z^3 \\
&= 1 + (-n - \ell - q + 6)z + (n\ell + nq + \ell q - 5n - 3\ell - q + 11)z^2 + (-n\ell q + 3n\ell - 6n + 6)z^3
\end{aligned}$$

which completes the proof.  $\square$

#### 4. SAGE PROGRAM

We have written a Sage program that computes the formula for the geodesic growth of a RACG with maximum clique size  $d$ . The setting for  $d$  can be changed by modifying in the second line of the program. For example, the program outputs

```

Numerator: -(1_1 - 3)*z + 2*z^2 + 1
Denominator: (1_0*1_1 - 2*1_0 + 2)*z^2 - (1_0 + 1_1 - 3)*z + 1

```

when  $d = 2$ , and

```

Numerator: (1_1*1_2 - 3*1_1 - 1_2 + 11)*z^2 + 6*z^3 - (1_1 + 1_2 - 6)*z + 1
Denominator: -(1_0*1_1*1_2 - 3*1_0*1_1 + 6*1_0 - 6)*z^3 + (1_0*1_1 + 1_0*1_2 +
1_1*1_2 - 5*1_0 - 3*1_1 - 1_2 + 11)*z^2 - (1_0 + 1_1 + 1_2 - 6)*z + 1

```

when  $d = 3$ . These formulas agree with the above results.

Here is the program text. We have also sent this program as a separate file.

```

1  # Clique size (CHANGE THIS)
2  d = 3
3
4
5  def generate_partitions(lower, upper, total):
6      """
7      Generator function that generates all partitions of the form
8          lower <= s_1 < t_1 <= ... <= t_n < s_n <= upper
9      such that (t_1 - s_1) + ... + (t_n - s_n) = total.
10     """
11
12     # Impossible case: yield no partitions
13     if upper - lower < total:
14         return
15
16     # Base case: yield empty partition
17     if total == 0:
18         yield []
19
20     # Chooses lower <= s_1 < t_1 <= upper such that t_1 - s_1 <= total.
21     # Then uses recursion to choose remaining s_i and t_i.
22     for s_1 in range(lower, upper - total + 1):
23         for t_1 in range(s_1 + 1, min(s_1 + total + 1, upper + 1)):
24             partitions = generate_partitions(
25                 t_1, upper, total - (t_1 - s_1))
26             for partition in partitions:
27                 yield [(s_1, t_1)] + partition
28
29
30 if __name__ == '__main__':
31     # Variables
32     z = var('z')
33     l = [var('l_' + str(i)) for i in range(0, d)]
34     l.append(0) # l[d] = 0
35
36     # N_{m, k}
37     N = {}
38     for m in range(1, d + 1):

```

```

39     N[(m, m)] = 1
40     N[(m, m - 1)] = l[m - 1] - l[m] - 1
41     for k in range(0, m - 1):
42         N[(m, k)] = prod([l[j] for j in range(k + 1, m)]) * sum([
43             binomial(m - k, j - k) * ((-1) ^ (j - k)) * l[j]
44             for j in range(k, m + 1)
45         ])
46
47     # M_{i, j}
48     M = {}
49     for i in range(0, d + 1):
50         for j in range(0, i + 1):
51             M[(i, j)] = 0
52
53             partitions = generate_partitions(j, d, i - j)
54             for partition in partitions:
55                 n = len(partition)
56                 M[(i, j)] += (-1) ** n * prod([
57                     binomial(t, s) for (s, t) in partition
58                 ]) * prod([
59                     N[(t, s)] for (s, t) in partition
60                 ])
61
62     # Geodesic growth
63     numerator = sum([
64         sum([
65             prod(l[:j]) * M[(i, j)]
66             for j in range(0, i + 1)
67         ]) * (z ** i)
68         for i in range(0, d + 1)
69     ])
70     numerator = numerator.expand().collect(z) # Simplify
71     denominator = sum([
72         M[(i, 0)] * (z ** i)
73         for i in range(0, d + 1)
74     ])
75     denominator = denominator.expand().collect(z) # Simplify
76

```

```
77     print('Numerator: ' + str(numerator))
78     print('Denominator: ' + str(denominator))
```

## REFERENCES

- [1] Yago Antolín and Laura Ciobanu, *Geodesic growth in right-angled and even Coxeter groups*, European Journal of Combinatorics **34** (2012), DOI 10.1016/j.ejc.2012.12.007.
- [2] Yago Antolín and Islam Foniqi, *Geodesic Growth of some 3-dimensional RACGs*, arXiv preprint arXiv:2105.09751 (2021).